
Schweizer Schreibweise: Semikolon, Punkt als Dezimalzeichen, Hochkomma als Tausender

11 Themen in der Reihenfolge des Lernpfads. Formel beginnt mit =, Argumente mit ; trennen, Text in
"Anführungszeichen".

[ ] = optionales Argument

01 Einfacher Zins
=Kapital*Zinssatz*Tage/360 (Bankusanz 360 Tage)
BEISPIEL A2 = 50'000, B2 = 0.0125, C2 = 90: =A2*B2*C2/360
→ 156.25
TYPISCHER FEHLER Zinssatz als 1.25 statt 1.25 % (0.0125)
eingegeben ergibt das Hundertfache.

02 Zinseszins
=Kapital*(1+Zins)^Jahre
BEISPIEL A2 = 25'000, B2 = 0.015, C2 = 8: =A2*(1+B2)^C2 →
28'162.31
TYPISCHER FEHLER Ohne Klammer um (1+B2) rechnet Excel
zuerst B2^C2.

03 Cashflow-Tabelle
=Saldo_Vorzeile+Cashflow (relativ, nach unten kopieren)
BEISPIEL C2 = 50'000 (Anfangssaldo), B3 = 12'000: =C2+B3
→ 62'000.00
TYPISCHER FEHLER Ein $ beim Saldo (=$C$2+B3) verhindert
die Fortschreibung beim Kopieren.

04 Schadenquote
=SUMME(Schäden)/SUMME(Prämien)
BEISPIEL Prämien B2:B5 = 10'000'000, Schäden C2:C5 =
7'108'500: =SUMME(C2:C5)/SUMME(B2:B5) → 0.711 = 71.1 %
TYPISCHER FEHLER Der Mittelwert der Einzelquoten
gewichtet kleine Sparten gleich stark wie grosse.

05 CAGR
=(Endwert/Startwert)^(1/Anzahl_Jahre)-1
BEISPIEL 2020: 84.2, 2025: 103.6 (A = Jahr, B = Wert):
=(B7/B2)^(1/(A7-A2))-1 → 0.0423 = 4.2 %
TYPISCHER FEHLER Anzahl Jahre ist die Differenz (5), nicht
die Anzahl Werte (6).

06 SVERWEIS ungefähr
=SVERWEIS(Suchkriterium;Matrix;Spaltenindex;WAHR)
BEISPIEL Staffel A2:B6 ab 0 | 100'000 | 250'000 …, E2 =
385'000: =E2*SVERWEIS(E2;$A$2:$B$6;2;WAHR) →
1'232.00 (Satz 0.0032)
TYPISCHER FEHLER Ist die erste Spalte nicht aufsteigend
sortiert, liefert WAHR falsche Sätze ohne Fehlermeldung.

07 SUMMENPRODUKT
=SUMMENPRODUKT(Array1;[Array2];…)
gewichtet: =SUMMENPRODUKT(Beträge;
Renditen)/SUMME(Beträge)
BEISPIEL Beträge B2:B5 = 400'000 total, Renditen C2:C5:
=SUMMENPRODUKT(B2:B5;C2:C5)/SUMME(B2:B5) → 0.0312
= 3.12 %
TYPISCHER FEHLER Der einfache MITTELWERT der Renditen
(3.075 %) ignoriert die Gewichte.

08 Break-even
=AUFRUNDEN(Fixkosten/(Preis-Variable_Kosten);0)
BEISPIEL A2 = 185'000, B2 = 49.00, C2 = 27.50:
=AUFRUNDEN(A2/(B2-C2);0) → 8'605 Stück
TYPISCHER FEHLER Ohne Klammer um (B2-C2) teilt Excel nur
durch den Preis.

09 RMZ
=RMZ(Zins;Zzr;Bw;[Zw];[F])
BEISPIEL A2 = 30'000, B2 = 0.049 p.a., C2 = 5 Jahre:
=-RMZ(B2/12;C2*12;A2) → 564.76 pro Monat
TYPISCHER FEHLER Jahreszins nicht durch 12 bzw. Jahre
nicht mal 12 gerechnet: Rate massiv falsch.

10 BW
=BW(Zins;Zzr;Rmz;[Zw];[F])
BEISPIEL A2 = 24'000 pro Jahr, B2 = 0.02, C2 = 15:
=BW(B2;C2;-A2) → 308'382.32
TYPISCHER FEHLER Zahlung positiv eingeben ergibt einen
negativen Barwert (Vorzeichen aus Sicht des Zahlenden).

11 ZW
=ZW(Zins;Zzr;Rmz;[Bw];[F])
BEISPIEL A2 = 7'258 pro Jahr, B2 = 0.015, C2 = 25:
=ZW(B2;C2;-A2) → 218'197.43
TYPISCHER FEHLER Einzahlungen zu Jahresbeginn brauchen
F = 1; ohne F rechnet Excel nachschüssig.

